



## Criando Extensões BASIC

O Calc pode converter todas as letras de um intervalo de células em maiúsculas. Mas, carece de um comando direto para transformar a primeira letra de cada palavra em maiúscula.

Neste artigo, apresentamos alguns conceitos sobre o ambiente BASIC, desenvolvendo uma solução simples para o problema acima. Os leitores interessados apenas na extensão, podem encontrá-la no portal <http://www.broffice.org>.




# TUTORIAL

## Criando a biblioteca

Uma biblioteca BASIC contém os módulos, diálogos e arquivos de configuração. Os últimos, criados automaticamente. Para criar a biblioteca, carregue o BrOffice.org e ative o diálogo Organizador de Macros, comandando:

- Ferramentas >> Macros >> Organizar Caixas de Diálogo
- Selecione a aba <Bibliotecas>



A lista *Biblioteca* exibe os nomes das bibliotecas existentes no *Local* selecionado, sendo que:

**Minhas macros e caixas de diálogo:** contém as bibliotecas armazenadas na área do usuário;

**Macros e caixas de diálogo do BrOffice.org:** contém as bibliotecas armazenadas na área compartilhada por todos os usuários;

**Documentos ativos:** contém as bibliotecas armazenadas dentro do documento selecionado.

Nossa biblioteca ficará na área do usuário, então:

- Na lista Local, selecione <Minhas macros e caixas de diálogo>;
- Clique sobre o botão de comando <Nova...>;
- No diálogo, digite o nome <TamLetra> para a biblioteca e clique sobre <Ok>.

Acabamos de criar a biblioteca. Mas, não feche o diálogo Organizador.

## Criando a interface gráfica


Precisamos de uma caixa de diálogo que ofereça algumas opções ao usuário, então:

- Na lista Biblioteca, selecione TamLetra;
- Clique sobre o botão Editar.



# TUTORIAL


Note que a janela do IDE Basic foi ativada, na sua parte inferior temos abas com os nomes dos módulos e diálogos existentes na biblioteca. Em nosso caso, apenas uma aba Module1.



## Para ativar a janela de edição de diálogos:

- Clique com o botão direito do mouse sobre a aba Module1;
- Selecione Inserir > Caixa de Diálogo do BASIC

Identifique os controles existentes na caixa de ferramentas, posicionando o ponteiro do mouse sobre os ícones. Em nosso diálogo vamos usar: dois botões de opção, linha horizontal, caixa de seleção, caixa de grupo e dois botões.



Após desenhar os controles no diálogo, altere a propriedade *Rótulo* de cada um deles conforme a figura acima (abra o diálogo Propriedades com um clique duplo sobre qualquer controle). Defina a propriedade *Estado* do primeiro botão de opção para *Selecionado*.

Defina a propriedade *Tipo do Botão* do botão Converter para *OK* e a do botão Cancelar para *Cancelar*.

Selecione o diálogo (clique sobre a borda) e altere a propriedade *Título*.

Após o desenho da interface gráfica, vamos ao código BASIC.

## O Código Fonte

Clique sobre a aba *Module1* para retornar ao editor de código, selecione e apague todo o seu conteúdo. Então, use o código fonte abaixo:

```
' Extensão que converte as letras iniciais das palavras  
' numa célula ou intervalo de células para maiúsculas.  
' Autor: Noelson A. Duarte, em Fev 2008.  
'
```

```
Sub ConverterLetrasIniciais  
  oDoc = thisComponent  
  oSelec = oDoc.getCurrentSelection()
```



# TUTORIAL

```
If (Not oSelec.supportsService("com.sun.star.sheet.SheetCellRange")) Then
    MsgBox "Selecione a célula ou o intervalo de células!", 48, "Erro"
Exit Sub
End If
DialogLibraries.loadLibrary ("TamLetra" )
oDialogo = CreateUNODialog (DialogLibraries.TamLetra.Dialog1)
iResp = oDialogo.execute()
If iResp = 0 Then
    Exit Sub
End If
bMinusc = False
If oDialogo.Model.CheckBox1.State = 1 Then
    bMinusc = True
End If
bNome = False
If oDialogo.Model.OptionButton2.State = 1 Then
    bNome = True
End If
LetrasIniciaisMaiusculas (oSelec, bNome, bMinusc)
End Sub

Sub LetrasIniciaisMaiusculas (oSel As Variant, bNomeProprio As Boolean, bMinuscula As Boolean)
vEnd = oSel.getRangeAddress()
nrLinhas = vEnd.EndRow - vEnd.StartRow
nrColunas = vEnd.EndColumn - vEnd.StartColumn
tipoTexto = com.sun.star.table.CellContentType.TEXT
For i = 0 To nrLinhas
    For j = 0 To nrColunas
        oCelula = oSel.getCellByPosition(j, i)
        If oCelula.getType() = tipoTexto Then
            sConteudo = oCelula.getString()
            If bMinuscula Then
                sConteudo = LCase(sConteudo)
            End If
            If bNomeProprio Then
                novoCont = LetrasNomeProprio(sConteudo)
            Else
                novoCont = TodasIniciaisMaiusculas(sConteudo)
            End If
            oCelula.setString(novoCont)
        End If
    Next j
Next i
End Sub

Function TodasIniciaisMaiusculas (sCadeia) As String
vPalavras = Split (sCadeia)
For i = 0 To UBound(vPalavras)
    letra = UCase(Left$(vPalavras(i),1))
    Mid(vPalavras(i), 1, 1, letra)
Next i
TodasIniciaisMaiusculas = Join(vPalavras())
End Function

Function LetrasNomeProprio (sCadeia) As String
vExcluir = Array("e","da","de","do","das","dos")
vPalavras = Split (sCadeia)
For i = 0 To UBound(vPalavras)
    bExcluir = False
    For j = 0 To UBound(vExcluir)
        If LCase(vPalavras(i)) = vExcluir(j) Then
            vPalavras(i) = LCase(vPalavras(i))
            bExcluir = True
            Exit For
        End If
    Next j
    If Not bExcluir Then
        letra = UCase(Left$(vPalavras(i),1))
        Mid(vPalavras(i), 1, 1, letra)
    End If
Next i
LetrasNomeProprio = Join(vPalavras())
End Function
```

IMAGEM



# TUTORIAL

## Vamos testar a nossa macro:

- Abra um novo documento do Calc;
- Preencha uma pequena extensão de células com texto;
- Selecione a extensão;
- Comande Ferramentas > Macros > Executar Macro;
- Expanda a entrada Minhas Macros;
- Expanda a biblioteca TamLetra e selecione Module1;
- Selecione a rotina principal ConverterLetrasIniciais;
- Comande Executar.

Recursos que facilitam a execução de uma tarefa beneficiam outros usuários. O BrOffice.org, via extensões, simplifica a distribuição e instalação de novos recursos. Vejamos como transformar a nossa macro numa extensão.

## Exportando como extensão

No BrOffice.org, uma extensão é um arquivo “zip” contendo código e arquivos correlatos. Se-guem os passos para criar uma extensão simples, a partir da nossa biblioteca Basic:

- Ferramentas > Macros > Organizar caixas de diálogo;
- Clique sobre a aba Bibliotecas;
- Selecione Minhas macros e caixas de diálogo;
- Selecione a biblioteca TamLetra e comande Exportar;
- No diálogo, marque Exportar como extensão e clique sobre Ok;
- No diálogo Exportar, defina a pasta e um nome (scTamLetra) para a extensão;
- Clique sobre o botão Salvar;
- Feche o diálogo Organizador de Macros.

Verifique, na pasta de destino, que um novo arquivo *scTamLetra.oxt* foi criado. Arquivos “.oxt” são identificados pelo sistema como Extensão do BrOffice.org.

Os usuários podem instalar a extensão através do Gerenciador de Extensão:

- Ferramentas > Gerenciador de Extensão ...

E, depois, usar o diálogo Seletor de Macros para executá-la:

- Ferramentas > Macros > Executar macro ...

Finalmente, a extensão desenvolvida ao longo do artigo, encontra-se no portal do BrOffice.org [1].

## Referência:

[1] <http://www.broffice.org/files/scTamLetra.oxt>